

CONOSCERE IL COMPUTER DIRETTAMENTE DAL COMPUTER

per Commodore 64

CONOSCERE
IL COMPUTER
DIRETTAMENTE
DAL COMPUTER


per Commodore 64



Beatrice d'Este

CONOSCERE
IL COMPUTER
DIRETTAMENTE
DAL COMPUTER

per Commodore 64



Beatrice d'Este



Beatrice d'Este

In questa lezione vediamo 5 comandi che dovranno essere usati in modo diretto sul listato del programma.

Comandi molto utili, che consentiranno ad esempio di mantenere contemporaneamente due programmi in memoria, fonderli insieme, ricercare delle stringhe o sostituirle all'interno del listato.

HMEM:

serve per trasferire il programma presente in memoria in una zona di sicurezza.

LMEM:

serve per richiamare il programma conservato con il comando HMEM.

APPEND:

serve per fondere insieme il programma in memoria con quello conservato con l'HMEM.

FIND:

serve per rintracciare tutte le linee di un programma che contengono una determinata stringa di caratteri. Dovrà essere usato nel modo: FIND.STRINGA., I-F.

CHANGE:

funziona in modo simile al FIND; sostituendo in più, con una nuova stringa, tutte quelle rintracciate e che si vogliono cambiare.

NB: Ogni volta che si vorranno utilizzare le istruzioni occorrerà prima caricare il programma HELP allegato alla lezione 31.

PROGRAMMIAMO INSIEME (CBM 64)

DATA BASE 3^a PARTE

```
12000 rem *** campo da totalizzare ***
12020 print"DATA BASE: Campo da totalizzare"
12070 K=0:K1=0
12075 K1=K1+1:if K1>n then 12130
12078 if t(K1)=1 then 12075
12090 K=K+1:x=2:y=3+(K+1)/2:gosub 34000:print right$(str$(K1),2); " -
"/c$(K1)
12100 if K1=n then 12140
12105 K1=K1+1:if K1>n then 12140
12108 if t(K1)=1 then 12105
```



```

12110 K=K+1:x=22:y=3+K/2:gosub34000:printright$(str$(K),2);" - ";
c$(K)
12120 ifK<nthen12075
12130 x=2:y=4+K/2:gosub34000:printright$(str$(n+1),2);" - TOTALE":
goto12200
12140 x=22:y=3+(K+1)/2:gosub34000:printright$(str$(n+1),2);" - TOT
ALE"
12200 x=2:y=20:gosub34000:input"Quale campo vuoi totalizzare"/z
12310 if(z<1)+(z>n+1)thengosub20000:goto12200
12320 ift(z)=1thengosub20000:goto12200
12400 return
14000 rem *** stampa generale ***
14020 print"DATA BASE: Stampa generale"
14040 input"Da quale codice";p:ifp=0thenreturn
14060 if(p<1)+(p>i)thengosub20000:goto14020
14080 x=4:y=18:gosub34000:print"Premi M per tornare al menu'"
14100 forh=ptoi
14120 open1,4,7
14200 print#1,"--- Codice: ";right$(" "+str$(h),3);"-----"
-----"
14220 forK=1ton
14240 print#1,right$(str$(K),2);
14260 ift(K)=0thenprint#1,">";c$(K);
14280 ift(K)=1thenprint#1,"-";c$(K);
14300 print#1,right$(" ",10-len(c$(K)));": ";d$(K,h)
14320 nextK
14340 print#1,"Totale: ";d$(n+1,h)
14500 getj$
14540 if(j$="m")+(j$="M")thenprint#1:close1:return
14600 print#1:close1:nexth
14700 return
16000 print"DATA BASE: Ordinamento"
16020 gosub23000:ifc=0thenreturn
16100 K=0
16120 K=K+1:x=2:y=3+(K+1)/2:gosub34000:printright$(str$(K),2);" -
";c$(K)
16130 ifK=nthen16160

```



```

16140 K=K+1:x=22:y=3+K/2:gosub34000:printright$(str$(K),2);" - ";c
$(K)
16145 ifK<nthen16120
16150 x=2:y=4+K/2:gosub34000:printright$(str$(n+1),2);" - TOTALE":
goto16200
16160 x=22:y=3+(K+1)/2:gosub34000:printright$(str$(n+1),2);" - TOT
ALE"
16200 x=11:y=20:gosub34000:input"Su quale campo";j
16220 if(j<1)+(j>n+1)thengosub20000:goto16200
16250 x=3:y=20:gosub34000:print"Ordinamento in corso, attendi ..."
16300 forx5=1toi-1:forx6=x5+1toi
16320 if(t(j)=0)*(val(d$(j,x6))>val(d$(j,x5)))then16500
16340 if(t(j)=1)*(d$(j,x6)>d$(j,x5))then16500
16360 forK=1ton+1
16380 bf$=d$(K,x5):d$(K,x5)=d$(K,x6):d$(K,x6)=bf$
16400 nextK
16500 nextx6,x5
16600 x=3:y=20:gosub34000:print"          Ordinamento completo          "
16620 gosub22000:return
17000 print"DATA BASE: Gestione file di dati"
17200 x=10:y=6:gosub34000:print"1 - Caricamento"
17220 x=10:y=8:gosub34000:print"2 - Registrazione"
17300 x=10:y=10:gosub34000:print"Scelta (1-2)"
17320 getgf$:if(gf$<"1")+(gf$>"2")then17320
17400 ifgf$="1"then19000
18000 rem *** registrazione dati ***
18020 print"DATA BASE: Registrazione dati"
18030 x=10:y=6:gosub34000:print"1 - Nastro"
18040 x=10:y=8:gosub34000:print"2 - Disco"
18050 x=10:y=10:gosub34000:print"Scelta (1-2)"
18060 getsu$:if(su$<"1")+(su$>"2")then18060
18070 ifsu$="1"thendv=1:nc=1:su$="nastro":no$=n$
18080 ifsu$="2"thendv=8:nc=2:no$="@0:"&n$&","&s$&","&w$&:su$="disco "
18090 print"DATA BASE: Registrazione su "&su$&""
18200 ifsu$="disco "thensu$="disco"
18240 x=4:y=6:gosub34000:print"Inserisci il "&su$&"" dei dati"
18250 gosub23000:ifc=0thenreturn
18270 x=0:y=10:gosub34000

```



```

18300 open1,dv,nc,no$
18310 print#1,rc
18330 print#1,i
18340 print#1,n
18350 print#1,n$
18400 forK=1ton
18410 print#1,c$(K)
18420 print#1,t(K)
18430 print#1,l(K)
18440 nextK
18500 forx=1toi
18520 forK=1ton+1
18530 q$=d$(K,x):ifq$=""thenq$="@"
18540 print#1,q$
18560 nextK,x
18580 close1
18600 return
19000 rem *** caricamento dati ***
19010 clr
19020 print"DATA BASE: Caricamento dati"
19030 x=10:y=6:gosub34000:print"1 - Nastro"
19040 x=10:y=8:gosub34000:print"2 - Disco"
19050 x=10:y=10:gosub34000:print"Scelta (1-2)"
19060 getsu$:if(su$<"1")+(su$>"2")then19060
19065 x=10:y=15:gosub34000:input"Nome file";n$
19066 if((su$="2")*(n$=""))+(len(n$)>16)then19065
19070 ifsu$="1"thendv=1:nc=0:no$=n$:su$="nastro":no$=n$
19080 ifsu$="2"thendv=8:nc=2:no$=n$+",s,r":su$="disco"
19090 print"DATA BASE: Caricamento da ";su$;" "
19200 ifsu$="disco"thensu$="disco"
19240 x=4:y=6:gosub34000:print"Inserisci il ";su$;" dei dati"
19250 ifdf=1thendf=0:gosub22000:goto19270
19260 gosub23000:ifc=0then4000
19270 x=0:y=10:gosub34000
19300 open1,dv,nc,no$
19310 input#1,rc
19330 input#1,i
19340 input#1,n
19350 input#1,n$
19360 dimc$(n+1),t(n+1),l(n+1),d$(n+1,rc)
19400 forK=1ton
19420 input#1,c$(K),t(K),l(K)
19440 nextK
19500 forx=1toi
19520 forK=1ton+1
19530 input#1,d$(K,x)
19540 ifd$(K,x)="@"thend$(K,x)=" "
19560 nextK,x
19580 close1
19600 goto4000

```


ISTRUZIONI PER L'UTILIZZO DEL DATA BASE.

Il programma serve per gestire un archivio di dati definito direttamente dall'utente.

Ad esempio, possono essere gestiti:

clienti	studi medici	fatture	dischi
fornitori	negozi	club	scadenze
magazzino	libri	entrate / uscite	contabilità

CARICAMENTO DEL PROGRAMMA DA CASSETTA:

accendere il computer, inserire la cassetta nel registratore, premere SHIFT e RUN/STOP, quindi il PLAY sul registratore e attendere.

INIZIO DEL PROGRAMMA:

dopo la presentazione iniziale premere ancora il RETURN per continuare. Il programma richiederà se si intende creare una nuova gestione.

Premere S per definire una nuova gestione e invece N per passare al caricamento dalla cassetta dei dati registrati in precedenza.

CREAZIONE NUOVA GESTIONE:

questa procedura è necessaria solo la prima volta che si usa il programma o quando si vuole creare una nuova gestione.

Inserire il nome della gestione che deve essere composto di sole lettere (massimo 16).

Inserire il numero dei campi desiderati per ogni record (massimo 20).

Per ogni campo bisognerà specificare:

- NOME CAMPO (massimo 10 caratteri)
- TIPO CAMPO (inserire A per definirlo alfanumerico e N per numerico)
- LUNGHEZZA CAMPO (massimo 24 caratteri se dichiarato A e invece 9 se dichiarato N)

Alla fine della definizione verrà presentato il numero di caratteri per record ed il numero massimo di records memorizzabili in un unico archivio.

MENU PRINCIPALE:

il menu principale permette le opzioni

- 1 - INSERIMENTO
- 2 - VARIAZIONE
- 3 - CANCELLAZIONE
- 4 - SITUAZIONE GENERALE
- 5 - RICERCA
- 6 - CAMPO DA TOTALIZZARE
- 7 - STAMPA
- 8 - ORDINAMENTO
- 9 - SAVE / LOAD FILE

INSERIMENTO:

inserire le informazioni in ogni campo. Dopo aver inserito tutte le informazioni verrà calcolato e visualizzato il totale automatico.

TOTALE AUTOMATICO:

il totale automatico, presente in ogni record, contiene la somma dei contenuti di tutti i campi numerici del record stesso.

Naturalmente viene calcolato automaticamente durante l'inserimento e la variazione dei dati. Viene visualizzato ogni volta che è mostrato il record.

RICONOSCIMENTO CAMPI NUMERICI ED ALFANUMERICI:

durante la visualizzazione di un record i campi dichiarati come numerici avranno il segno del maggiore (>) subito dopo il numero del campo.

SUPERAMENTO CAPACITA' DI MEMORIA:

nel caso si cerchi di inserire più records di quelli calcolati in fase di definizione, il programma visualizzerà il messaggio indicante il superamento delle capacità di memoria. Si consiglia di salvare il file di dati su cassetta.

VARIAZIONE:

inserire il numero del record da variare. Quindi il numero del campo da variare e correggerlo. Inserire lo Ø per terminare e tornare al menu.

CANCELLAZIONE:

inserire il numero del record da cancellare. Confermare con S la cancellazione. L'ultimo record inserito prenderà il posto (e quindi il codice) di quello cancellato.

RISPOSTE DI CONFERMA:

a volte il programma richiede una conferma. Premere la S per confermare oppure la N per non confermare.

SITUAZIONE GENERALE:

inserire il codice del record da cui far partire la visualizzazione. Durante la visualizzazione è possibile usare i tasti:

- A - per avanzare al record successivo
- I - per tornare al record precedente
- M - per tornare al menu.

RICERCA:

la ricerca permette di ritrovare un certo record o una serie di records. Scegliere su quale campo operare la ricerca (si può ricercare su tutti i campi definiti).

Scegliere il tipo di ricerca per informare il programma del tipo di indirizzo da inserire:

- INCOMPLETO - se l'indizio comprende solo i primi caratteri di quello cercato
- IDENTICO - se l'indizio è uguale a quello cercato
- INTERNO - se l'indizio è interno a quello cercato
- MAGGIORE - se l'indizio è maggiore di quello cercato
- MINORE - se l'indizio è minore di quello cercato.

Inserire l'indizio da ricercare e battere il RETURN.

Durante la visualizzazione dei records trovati verrà mostrato il TOTALE AUTOMATICO e la SOMMA dei contenuti del CAMPO DA TOTALIZZARE.

CAMPO DA TOTALIZZARE:

permette di scegliere su quale campo numerico operare la totalizzazione durante la ricerca.

STAMPA:

l'opzione è identica a quella della SITUAZIONE GENERALE, con la differenza che avviene la stampa su stampante.

ORDINAMENTO:

permette di ordinare (alfabeticamente o numericamente, in base a qualsiasi campo) i dati dell'intero archivio. Tieni presente che se ci sono parecchi dati questa operazione potrà richiedere molto tempo, anche alcune ore.

REGISTRAZIONE E LETTURA DATI:

registra e carica i dati dalla cassetta in modo da conservarli quando il computer viene spento.

RITORNO AL PROGRAMMA SENZA LA PERDITA DEI DATI:

se si interrompe l'esecuzione del programma con RUN/STOP sarà possibile ripartire senza alcuna perdita di dati scrivendo GOTO 4ØØØ.